
1. Vorwort

Diese  Dokumentation  beschreibt  den  Einsatz  des  Kleinplotters  XY 4131  am 
Kleincomputer KC 85/3-5 von Mühlhausen mithilfe des Moduls M001 und einer 
Erweiterung des BASIC-Interpreters.

Als Vorlage diente das Programm GRPLOT und die Bedienungsanleitung zum 
KLEINPLOTTER XY 4131 am robotron Z9001, KC 85/1 bzw. KC 87. Vielen Dank 
an Volker Pohlers für die Bereitstellung der Handbuch-Vorlage und der Quelltexte 
von GRPLOT.

Vielen Dank auch an Wolf-Peter Wruck für die Bereitstellung der Plotter-Software 
vom KC 85/3, welche die Basis des hier beschriebenen Plotter BASIC bildet.

Impressum der Bedienungsanleitung von robotron:

1/87 b
VEB ROBOTRON - MESSELEKTRONIK "OTTO SCHÖN" DRESDEN
Lingnerallee 3, Postschliessfach 211, Dresden, DDR – 8012

LABORATORNÍ PŘÍSTROJE, k.p. PRAHA
Praha 6, Na okraji 335, ČSSR

Průvodní technická dokumentace

Edice Zapisovače, míchačky, termostaty

Ediční číslo ZM-031-026-87
Formát A5
Počet stran 40
Náklad 1.500 výtisků
1. vydání
Obálku navrhl Pavel Čampulka
Redakce Jana Příchystalová
Technická redakce a příprava výroby Jana Hodkováá
Vydaly Laboratorní přístroje, k.p., Praha, 1987
Vyrobil Mír, novinářské závody, n.p., Praha, 1987

1.1. Entstehungsgeschichte

Im Januar 2022 wurde von Wolf-Peter Wruck im robotrontechnik-Forum eine Dis-
kussion zum Einsatz der tschechischen Kleinplotter XY 41xx am KC 85/3 gestar-
tet. Er hatte seinerzeit zwei Nachnutzungen für die Anbindung des Plotters am 
KC 85/3 erworben.

• Eine  Variante  war  das  BASIC-Programm  PLOTTER.SSS  von  W. 
Domschke aus Mühlhausen. Am Anfang dieses BASIC-Programms ist 
ein Maschinencode-Teil mit den Treiber-Routinen für den Plotter einge-
bettet. Man arbeitet also in BASIC und hat Unterprogramme, welche die 
einzelnen Plotterbefehle realisieren. Das Modul M001 kann sich dabei in 
einem beliebigen Steckplatz befinden.

◦ GOSUB 20 Initialisierung Plotter
◦ GOSUB 30 Linie zeichnen
◦ GOSUB 40 zu einem Punkt gehen ohne zu zeichnen
◦ GOSUB 50 Kreis zeichnen
◦ GOSUB 60 Kreisbogen zeichnen
◦ GOSUB 70 Umschaltung Plotter / Bildschirm
◦ GOSUB 80 Einstellung Linientyp

• Bei der zweiten Variante handelt es sich um die Portierung der KC 87-
Software GRPLOT für den KC 85/3. Hier wird ein kleines Maschinen-
code-Programm unter CAOS geladen und gestartet. Es muss sich ein 
Modul M001 im Schacht 08 und ein Modul M011 im Schacht 0C befin-
den. Nach dem Start werden die Module überprüft, das M011 auf 4000H 
geschaltet, der Inhalt des BASIC-ROM auf 4000H kopiert, die erweiter-
ten BASIC Befehle in den RAM kopiert und die Ansprungtabelle modifi-
ziert. Danach wird der 16K-Block mit dem geänderten BASIC auf C000H 
geschaltet,  BASIC angesprungen  und der  maximale  BASIC Speicher 
begrenzt, weil am Ende des BASIC-Speichers die neuen Plotterbefehle 
liegen. Nun kann man den Plotter mit Befehlen ähnlich wie beim KC 87 
ansprechen.

Die beiden Varianten unterscheiden sich zudem noch in der verwendeten Hard-
ware. Während der Plotter in der Mühlhausener Variante an Port B des M001-
PIO anzuschließen ist,  verwendet  die zweite Variante den PIO-Port  A mit  der 
Anschlussbelegung wie beim KC 87. Um beide Programme nutzen zu können, 
sind also 2 verschiedene Kabel erforderlich!

Die zweite Variante weckte mein Interesse. Zunächst interessierte mich mangels 
Plotter allerdings hauptsächlich die Technik, wie so eine BASIC-Erweiterung pro-
grammiert  wurde.  Denn  schließlich  befinden sich  sowohl  der  BASIC-Kern  als 
auch die BASIC-Erweiterungen komplett im ROM und lassen sich nicht ohne wei-
teres ändern.

Die Plotter-Erweiterung bringt 10 neue BASIC-Befehle mit. Da beim BASIC des 
KC 85 (im Gegensatz zum KC 87) die Befehle LINE, CIRCLE und PSET bereits 
existieren, wurde den Plotter-Befehlen zur Unterscheidung ein  P vorangestellt. 
Die restlichen Befehle sind identisch zum KC 87.

KC     87  KC     85  

LINE PLINE

CIRCLE PCIRCLE

SCREEN SCREEN

ZERO ZERO

HOME HOME

PSET PPSET

SIZE SIZE

LABEL LABEL

SCALE SCALE

GCLS GCLS

Bild 1: Vergleich Plotter-BASIC-Befehle KC 87 und KC 85 (Plotter BASIC 1.0)

Wolf-Peter stellte mir alle Programme freundlicherweise zur Verfügung, sodass 
ich  diese  untersuchen  konnte.  Erhalten  hatte  ich  zunächst  den  Treiber,  ein 
Assemblerlisting  der  für  CAOS  4.1  modifizierten  Laderoutine  sowie  ein  paar 
handschriftliche Notizen. Später habe ich dann noch das Original-INIT-Programm 
für Kassette bekommen, mit dem ich den Rest der Modifikationen klären konnte.

Um Anpassungen vornehmen zu können, musste der Plotter-Treiber zunächst 
reassembliert  werden.  Hier  stellte  es  sich  als  hilfreich  heraus,  dass  Volker 
Pohlers  auf  seiner  Website  bereits  ein  zum  Teil  kommentiertes  Listing  von 
GRPLOT  des  KC 87  veröffentlicht  hatte.  Diesen  Teil  konnte  ich  fast  1:1 
übernehmen.  Ich  habe  dabei  auch  noch  weitere  Reste  KC 87-spezifischen 
Programmcode  gefunden,  welche  bei  der  Portierung  für  den  KC 85/3  nicht 
entfernt worden sind.

Entwickler für die Version KC 85/1, KC 87 und Z9001:
VEB Robotron-Meßelektronik „Otto Schön“ Dresden

Anpassung für den KC 85/3 mit CAOS 3.1:
Betriebsberufsschule „Rudolf Lattner“
VEB Kombinat LACKE und FARBEN Dresden

Weiterentwicklung für den KC 85/3 bis KC 85/5:
Mario Leubner, KC-Club, 2022

1.2. Plotter BASIC 1.0

Zunächst  sollte  eine  einheitliche  Anschlussbelegung für  den Plotter  am M001 
definiert werden. Die Entscheidung ist zur Mühlhausener Variante gefallen, wel-
che der Empfehlung von W. Domschke in rfe 4/1989 folgt und in den Bildern  5 
und 6 dargestellt ist.

BASIC-Befehle werden intern als Token verarbeitet, das sind Bytes im Bereich 
zwischen 80H und FFH. Unter CAOS 3.1 bzw. CAOS 4.2 sind die Token 80H bis 
F4H mit BASIC-Worten belegt. Die Plotter-Erweiterung nutzt 10 neue Token von 
F5H bis FEH. Ab CAOS 4.5 wurden jedoch bereits neue BASIC-Token für den 
Laufwerkswechsel bzw. ab CAOS 4.7 für die Device-Funktionen definiert, eben-
falls ab F5H. Das heißt, nach dem Laden des Plotter-BASIC stehen diese Device-
Funktionen nicht mehr zur Verfügung. Im Bedarfsfall muss man also ins CAOS-
Menü wechseln und von dort aus das Laufwerk bzw. Device ändern.

Ein zweiter Nachteil war die fest vorgegebene Modulbestückung. Viele Anwender 
haben auf den Steckplätzen 08 und 0C bereits Module gesteckt und müssten für 
den Betrieb des Plotters die Module umstecken.

Das dritte Problem ist die Art und Weise, wie die Sprungadressen der Plotter-
Erweiterung  eingebunden  wurden:  Es  wird  eine  vorbereitete  Sprungtabelle 
benutzt, welche eine Kopie der Sprungtabelle von CAOS 3.1 darstellt  und am 
Ende um die zusätzlichen Plotter-Befehle ergänzt wurde. Damit läuft  das Pro-
gramm nur unter CAOS 3.1.

Mit  der ersten Version meines Plotter-BASIC habe ich folgende Anpassungen 
vorgenommen:

• Die Plotter-Schnittstelle wurde auf PIO-Port B des Moduls M001 umge-
stellt.

• Die benötigten Module M001 und M011 können sich in beliebigen Steck-
plätzen befinden.

• Für den KC 85/4 ist auch ein 16K-RAM-Modul M022 als Speichererwei-
terung ausreichend, da der KC 85/4 bereits über RAM auf den Adressen 
4000H bis BFFFH (RAM4, RAM8) verfügt.

• Die Sprungadressen der Erweiterungstoken von 80H bis F4H werden 
aus dem ROM der jeweiligen CAOS-Version übernommen, unter wel-
cher das Plotter-BASIC gestartet wird.

• Es wird kein separates INIT-Programm mehr benötigt, die Initialisierung 
ist nun als Startroutine im Hauptprogramm enthalten. Damit kann das 
Plotter-BASIC von einem beliebigen Datenträger geladen werden.

Die  erste  Version  des  Plotter-BASIC läuft  auf  dem KC 85/3  bis  KC 85/5  und 
wurde unter  CAOS 3.1 bis  CAOS 4.8 getestet.  Beim KC 85/3 muss vor  dem 
Laden des Plotter-BASIC das RAM-Modul eingeschaltet  werden, das das Pro-
gramm in den RAM4 eingelesen wird.

Der KC 87 benutzt intern andere Token als der KC 85. Die Übernahme von Pro-
grammen  funktioniert  nur  mit  einem  LIST#1  am  KC 87  und  anschließendem 
LOAD#1 am KC 85, also mittels Programmlisting im Klartext.  Zu Beachten ist 
dabei noch die Besonderheit, dass die drei Befehle LINE, CIRCLE und PSET bei 
diesem Vorgang entsprechend Tabelle in Bild 1 umgeschrieben werden müssen 
in PLINE, PCIRCLE und PPSET.

Mario Leubner, 02.10.2022

1.3. Plotter BASIC 2.0

Um das mühevolle Umschreiben der Befehle LINE, CIRCLE und PSET zu umge-
hen, habe ich mir schließlich einen Trick ausgedacht. Der Plotter-Befehl SCREEN 
aktiviert beim KC 87 die Plotter-Schnittstelle und schaltet zwischen dem Grafik-
Bildschirm und dem Plotter als Ausgabegerät um. Grafische Ausgaben können 
also entweder  auf  dem Bildschirm oder  auf  dem Plotter  ausgegeben werden. 
Beim KC 85 wurde vom Befehl SCREEN im Plotter BASIC 1.0 bisher nur die Initi -
alisierung  der  Plotter-Schnittstelle  vorgenommen.  Einen  separaten  Grafik-Bild-
schirm wie beim KC 87 gibt es nicht, jedoch Grafik-Befehle für die Darstellung auf 
dem „normalen“ Bildschirm des KC 85. Was liegt also näher, als dass man auch 
beim  KC 85  die  Grafik-Befehle  bei  Aktivierung/Deaktivierung  des  Plotters 
umschaltet. Das bringt zwei wesentliche Vorteile:

• Die Befehle LINE, CIRCLE und PSET müssen nicht mehr umgeschrie-
ben werden bei der Übernahme von KC 87-Plotter-Programmen mittels 
LIST#1 und LOAD#1.

• Statt 10 werden nur noch 7 neue Token benötigt und die Befehle zur 
Laufwerks- bzw. Device-Umschaltung stehen weiter zur Verfügung.

Aufpassen muss man allerdings bei der gleichzeitigen Verwendung der drei Gra-
fikbefehle in einem Programm, da diese für Plotter und Bildschirm nun eine unter-
schiedliche Syntax aufweisen.

Plotter-BASIC 2.0 wird in den RAM0 ab der Adresse 300H eingelesen. Das vor-
herige Zuschalten eines RAM-Moduls beim KC 85/3 ist damit auch nicht  mehr 
erforderlich.

Die weiteren Aussagen in diesem Handbuch beziehen sich auf die Version 2.0 
des Plotter-BASIC.

Mario Leubner, 23.12.2022

2. Verwendungszweck

Der Kleinplotter XY 4131 (GRAFICKÁ JEDNOTKA XY 4131) vom Hersteller

LABORATORNÍ PŘÍSTROJE
162 03 Praha 6/ČSSR

ist ein leistungsfähiges grafisches Ausgabegerät für die Kleincomputer robotron 
Z 9001, KC 85/1 und KC 87 und jetzt auch für die Kleincomputer KC 85/3-5 aus 
Mühlhausen. Mit dem Plotter lassen sich grafische Darstellungen und Texte im 
Format DIN A4 zeichnen.

In dieser Anleitung wird der Einsatz des Kleinplotters XY 4131 beschrieben. In 
der Reihe der tschechischen Kleinplotter gibt es außer dem XY 4131 auch noch 
einen XY 4140 sowie einen XY 4150. Die beschriebene Hardware und Software 
wurde mit dem XY 4131 getestet. Die anderen Plotter dieser Baureihe sind funkti-
onskompatibel können genauso eingesetzt werden. Unterschiede sind gegebe-
nenfalls den jeweiligen Original-Dokumentationen der Plotter zu entnehmen.

Bild 3 zeigt die Grundausstattung.

Bild 3: Grundausstattung

1 Kleinplotter XY 4131 4 Modul M001
2 Anschlusskabel (Computeranschluss) 5 Bedienungsanleitung
3 Programmkassette 6 Plotterstifte

Die Lage der Bedienelemente ist im Bild 4 dargestellt.

Bild 4: Bedienelemente des Kleinplotters XY 4131

7 Buchse für Computeranschluss 12 Bereit-Anzeige
8 Netzkabel 13 Stiftfeineinstellung
9 Hebel für Papierhalterung 14 Stifthalterung
10 Netzschalter/Anzeige 15 Papierbahn
11 Taste für Stift heben/senken 16 Andruck- und Transportrollen

13



Plotter BASIC 2.0 für XY 4131

3. Lieferumfang und technische Daten

Die Ausführungen zu Lieferumfang und Ersatzteilen wurden 1:1 aus dem robo-
tron-Handbuch übernommen und sind nur informativ anzusehen. Das Plotter-Gra-
fik-Modul sowie die Programmkassette R0114 sind für den Einsatz am KC 85/3-5 
nicht  erforderlich.  Das  Anschlusskabel  für  den  KC 85 muss  selbst  angefertigt 
bzw. ein vom KC 87 vorhandenes Kabel umgebaut werden.

3.1. Lieferumfang

Zum Lieferumfang gehörte:

1 Kleinplotter XY 4131
1 Plotter-Grafik-Modul 690 033.2; ZAK-Nr. 138 21 78 005 000435
1 Programmkassette R0114 536 647.3; ZAK-Nr. 138 21 78 005 000427
1 Anschlusskabel (Plotter/Computer) 536 648.1
1 Geräteschutzhülle 40 496
1 PVC-Beutel 38 141
5 Schmelzsicherungen T80mA/250V 24 039
10 Plotterstifte KIN 0577 – blau 42 751
10 Plotterstifte KIN 0577 – rot 42 752
10 Plotterstifte KIN 0577 – schwarz 42 753
10 Plotterstifte KIN 0577 – grün 42 754
10 Plotterstifte KIN 0577 – violett 42 755
1 Bedienungsanleitung
1 Garantieurkunde für den Kleinplotter XY 4131
1 Garantieurkunde für den Plotter-Grafik-Modul

3.2. Technische Daten XY 4131

Papierformat DIN A4
Effektive Zeichenfläche 255 mm x 180 mm
Raster 2550 x 1800 Schritte
Minimale Schrittweite 0,1 mm
Zeichengeschwindigkeit 100 mm/s
Zeichensatz ASCII und Sonderzeichen
Schriftart
- Zeichengröße Programmierbar

(Standard: Höhe 4 mm)
- Zeichenform Programmierbar

(Standard: Höhe/Breite = 5/3)
- Schriftneigung Programmierbar

(Standard: senkrecht)
Plotterstift-Steuerung
(Heben und Senken)

Programmierbar oder mit Schalter

Plotterstift-Nullstellung Programmierbar
(Standard: links unten)

Plotterstifte System FIX - 5 Farben (KIN 0577), Tuschezei-
chengeräte markant (Lin's 9 Plus), Faserschreiber

Papier Format DIN A4, 60 bis 80 g/m²
Antrieb 2 Schrittmotoren

X-Richtung Papiertransport
Y-Richtung Stiftbewegung

Stromversorgung
- Netzspannung 220 V + 22 V
- Netzfrequenz 50 bis 60 Hz
- Leistungsaufnahme 14 VA
Schutzklasse 1
Abmessungen 375 mm x 255 mm x 110 mm
Masse 5 kg

4. Wirkungsweise

4.1. Hardware

Zwischen dem Kleincomputer und dem Kleinplotter XY 4131 existiert eine geräte-
spezifische Schnittstelle. Die Steuersignale für den Kleinplotter werden über die 
E/A-Buchse des Moduls M001 ausgegeben.
Entsprechend den vom Kleincomputer  angelegten  Signalen  werden durch  die 
Steuerplatine des Kleinplotters zwei Schrittmotoren angesteuert. Die Bewegung 
in X-Richtung erfolgt durch Drehen der Rollen der Papierhalterung und die Bewe-
gung in Y-Richtung durch den Transport der Stifthalterung. Außerdem wird das 
Heben und Senken des Plotterstiftes gesteuert.

4.2. Software

Zum Betreiben des Kleinplotters am KC 85/3-5 werden benötigt:

• ein Modul M001 DIGITAL IN-OUT mit angeschlossenem Plotter
• RAM im Adressbereich 4000H bis BFFFH
• das Maschinencodeprogramm PLBASIC.KCC (Plotter BASIC 2.0)

Das Modul M001 kann in einem beliebigen Modulschacht gesteckt sein. Das Pro-
gramm ist selbststartend und sucht, beginnend ab Steckplatz 8 das erste M001 
und schaltet es ein. Befinden sich mehrere Module M001 im KC-System, so ist 
der Plotter am Modul auf dem niedrigsten Steckplatz anzuschließen.

Plotter-BASIC testet bei Programmstart, ob im Adressbereich 4000H bis BFFFH 
bereits RAM eingeblendet ist. Ist das der Fall, wird dieser RAM direkt benutzt. So 
ergibt sich die Möglichkeit, beliebige RAM-Module zu verwenden. Diese sind vor 
dem Laden des Plotter-BASIC manuell  einzuschalten. Falls kein RAM manuell 
aktiviert  wurde,  sucht  das  Programm nach einem 64K-RAM-Modul  M011 und 
schaltet dieses passend ein. Beim KC 85/4-5 ist auch ein 16K-RAM-Modul M022 
für die Speicherlücke ab C000H ausreichend. Der RAM4 und RAM8 darf dabei 
nicht abgeschaltet sein.

Werden die benötigten Module nicht vorgefunden, dann bricht das Programm mit 
einer entsprechenden Fehlermeldung ab.

Plotter-BASIC 2.0 erweitert das im ROM des KC 85/3-5 enthaltene BASIC um 
zusätzlichen Grafikanweisungen zum Steuern des Kleinplotters.

5. Anschluss und Inbetriebnahme

Der Kleincomputer, das Fernsehgerät und der Kassettenrecorder sind entspre-
chend der Bedienungsanleitung des Kleincomputers aufzustellen und anzuschlie-
ßen.

5.1. Stecken und Entfernen von Erweiterungsmodulen

Stecken Sie ein Modul M001 DIGITAL IN/OUT und ein 64K RAM-Modul M011 
(oder ein 16K RAM-Modul M022 beim KC 85/4-5) in beliebige Modulschächte des 
Kleincomputers. Die Modulprioritätskette muss dabei immer geschlossen bleiben. 
Also sind zuerst im Grundgerät der Steckplatz 8 (rechts), danach der Steckplatz 
C (links) und anschließend weitere Steckplätze von Erweiterungsaufsätzen in auf-
steigender Reihenfolge zu belegen.

Achtung!

Das Stecken und Entfernen von Modulen darf nur im ausgeschalteten Zustand 
des KC 85-Systems vorgenommen werden!
Damit ergeben sich folgende Handgriffe für das Stecken von Modulen:

a) Den Computer bzw. Aufsatz ausschalten.

b) Die Kappe des Modulschachtes abnehmen. Hierzu muss die Kappe an 
den gegenüberliegenden Griffflächen leicht zusammengedrückt werden.

c) Das Modul bis zum fühlbaren Einrasten einschieben. Das Modul ist dann 
richtig kontaktiert, wenn sein hervorstehender Rand unmittelbar an der 
Gerätewand anliegt.

d) Nun kann der Computer und ggf. Aufsatz wieder eingeschaltet werden.

Bevor die weitere Bedienung beschrieben wird, soll an dieser Stelle gleich das 
Entfernen von Modulen aus dem System erläutert werden. Zum Entfernen eines 
Moduls sind folgende Schritte notwendig:

a) Den Computer bzw. Aufsatz ausschalten.

b) Den linken und den rechten Zeigefinger unter den Modulkopf legen und 
mit den Daumen die seitlich am Modul befindlichen Hebel gleichzeitig 
nach unten drücken. Dabei rastet das Modul aus und wird etwa einen 
Zentimeter  aus  dem Gerät  heraus  geschoben.  Nun kann das  Modul 
leicht aus dem Schacht gezogen werden.

c) Die Kappe auf die Schachtöffnung stecken.

5.2. Anschluss und Programmstart

Zum Anschluss des Kleinplotters XY 4131 wird ein Anschlusskabel benötigt, wel-
ches gemäß Schaltplan Bild 5 bzw. Tabelle Bild 6 anzufertigen ist.

18

Plotter Signal Modul M001 PIO-Port

1 PEN 9B B7

6 Masse 1A+1B

3 X-Y 3B B1

4 V-R 4B B2

5 /READY 7B
8B (Diode zu 7B)

B5
B6

2 STEP 2B B0
Bild 6: Signalbelegung Plotterbuchse und Modul M001 nach rfe 4/89

Mit dem Anschlusskabel ist die Buchse für den Computeranschluss (7) des Klein-
plotters mit der Buchse des Moduls M001 am Computers zu verbinden. Schalten 
Sie nach dem Stecken der Module und des Plotter-Anschlusskabels alle Geräte 
ein.

Laden  Sie  nun  vom  CAOS-Menü  aus  das  Maschinencodeprogramm 
PLBASIC.KCC z. B. durch

%LOAD  ENTER 

Nach dem Laden des Programms wird der BASIC-Interpreter automatisch gestar-
tet. Er meldet sich mit

Plotter XY41xx BASIC 2.0

MEMORY END ? :

Beantworten Sie die Frage mit  ENTER  oder einem Wert bis maximal

43246

Danach ist der BASIC-Interpreter mit zusätzlichen Grafikanweisungen arbeitsbe-
reit. Folgende 7 zusätzliche BASIC-Befehle stehen nun zur Verfügung:

SCREEN, ZERO, HOME, SIZE, LABEL, SCALE, GCLS

Die im KC 85-BASIC bereits enthaltenen 3 Befehle

PSET, LINE, CIRCLE

erhalten im Plotter-Modus eine abweichende Syntax.

5.3. Papier- und Stifthalterung

Durch  ein  Absenken  des  Hebels  für  die  Papierhalterung  (9)  heben  Sie  die 
Andruckrollen (16) und können nun Ihr Papier (A4) oder eine Folie im entspre-
chenden Format (bei Verwendung geeigneter Stifte) in die Papierbahn (15) einle-
gen und durch Anheben des Hebels (9) festklemmen. Achten Sie darauf, dass 
das Papier parallel zur Papierbahn liegt, da es während des Zeichnens mithilfe 
der Transportrollen verschoben wird.

In die Stifthalterung (14) können Sie nun einen geeigneten Zeichenstift einklem-
men und durch Links- oder Rechtsdrehen in die notwendige Höhe positionieren. 
Richtig positioniert ist der Stift, wenn sich die Schreibspitze etwa 2 mm über dem 
Zeichenpapier befindet. Eine Feineinstellung können Sie mithilfe des Hebels (13) 
vornehmen.

5.4. Papier und Stifte

Als Zeichenpapier für den Plotter können Sie jedes Schreib- oder Zeichenpapier 
mit einer glatten, festen Oberfläche und einer Masse von 60 bis 80 g/m² verwen-
den. Ebenso können Sie auf genügend steife Folien zeichnen.

Als Ersatz für die nicht mehr erhältlichen Plotterstifte eignen sich Faserschreiber 
und Fineliner. Sie sollten nicht länger als 7 cm und etwa 10 mm stark sein sowie 
eine feste Schreibspitze besitzen. Vorteilhaft sind Rillen oder eine rauhe Oberflä-
che im Klemmbereich,  um ein  Verrutschen des  Stifts  zu  verhindern.  Bewährt 
haben  sich  bei  mir  zum  Beispiel  der  „STAEDTLER  pigment  liner“,  und  der 
„OFFICE Liquid Longliner“, siehe Bild 7.

Sie  können  aber  auch  geeignete  Tuschezeichengeräte  verwenden,  wenn  Sie 
saubere Zeichnungen auf  Papier  oder  Folie  anfertigen wollen.  Geeignet  dazu 
sind z. B. die Tuschezeichengeräte markant (Lin's 9 Plus) bei Verwendung der 
Zeichentusche (HSL 57841126).

5.5. Beispiel

Nachdem Sie den Kleinplotter gemäß Abschnitt 5.1. bis 5.3 zur Nutzung vorberei-
tet haben, können Sie die Funktionsweise durch das Programm

10 SCREEN 0,1

20 ZERO(0,0):HOME

30 LINE(0,0)-(2550,1800),1,B

40 PSET(700,700),0

50 LABEL "KC 85 und XY 4131"

60 HOME

testen. Dargestellt wird ein Rechteck mit der Schrift:

KC 85 und XY 4131

Ebenso steht Ihnen das BASIC-Demonstrationsprogramm "PLDEMO.SSS" zum 
Funktionstest des Plotters zur Verfügung.

6. Grafische Anweisungen (BASIC)

Die nachfolgend beschriebenen BASIC-Anweisungen gelten für  die  grafischen 
Ausgaben auf den Kleinplotter XY 4131. Die Auflösung beträgt für dieses gra-
fische Ausgabegerät  2550 x 1800 Punkte  à  0,1 mm. Jeder  dieser  Punkte  des 
Gerätekoordinatensystems  kann  mit  den  entsprechenden  Anweisungen  ange-
sprochen werden.

Gerätekoordinatensystem:

(0, 1800) (2550,1800)

(2550, 0)
(0, 0)

Hinweis:
Wurde das Programm PLBASIC.KCC nicht  geladen,  so führen alle grafischen 
Anweisungen zu einem Syntax-Fehler (?SN ERROR).

6.1. Grundeinstellungen des Plotters

SCREEN Zuweisung des Plotters
ZERO Festlegung des Koordinatenursprungs
SCALE Skalierungsfaktoren
HOME Stift zum Koordinatenursprung
GCLS Grundeinstellung des Plotters

Bevor die grafischen Anweisungen zum Zeichnen und zur Schriftausgabe in der 
gewünschten Weise wirken, müssen Sie den Plotter als grafisches Ausgabegerät 
zuweisen und, falls erforderlich, Ihr aktuelles Koordinatensystem definieren.

Zuweisung des Plotters

Format:
SCREEN [0],plotter

plotter - Parameter, der den Plotteranschluss spezifiziert
0 - Plotter aus
1 - Plotter ein

Funktion:
Mit plotter = 1 wirken alle nachfolgenden Grafik-Anweisungen auf den Plotter als 
Ausgabegerät. Die BASIC-Befehle PSET, LINE und CIECLE wirken nun auf den 
Plotter entsprechend der in dieser Dokumentation angegebenen Syntax. Mit plot-
ter = 0 wird der Plotter als Ausgabegerät abgeschaltet. Die BASIC-Befehle PSET, 
LINE und CIRCLE wirken wieder auf den Bildschirm entsprechend der im BASIC-
Handbuch beschriebenen Syntax.

Hinweis:
Mit  der  ersten  SCREEN-Anweisung  werden  die  Skalierungsfaktoren  entspre-
chend der Anweisung SCALE 1,1 initialisiert.

Solange keine SCREEN-Anweisung abgearbeitet ist, sind die Skalierungsfakto-
ren nicht initialisiert und damit undefiniert. Bei Anweisungen, die eine Stiftpositio-
nierung enthalten, kann das zum ?OV ERROR führen.

Nullpunkt festlegen

Format:
ZERO(x,y)

Funktion:
Der Nullpunkt des aktuellen Koordinatensystems wird in den Punkt (x,y) gelegt.

Hinweis:
1. Bezugspunkt ist der Nullpunkt des Gerätekoordinatensystems (links unten).
2. Die Koordinatenangaben werden mit den durch SCALE eingestellten Skalie-

rungsfaktoren multipliziert.

Festlegen der Maßstäbe

Format:
SCALE xfaktor, yfaktor
xfaktor - Skalierungsfaktor für x-Richtung
yfaktor - Skalierungsfaktor für y-Richtung

Funktion:
Mit SCALE erfolgt eine Maßstabsfestlegung in x- bzw. y-Richtung (unabhängig 
voneinander).
Alle Koordinatenangaben werden vor der Verarbeitung mit xfaktor bzw. yfaktor 
multipliziert.

Hinweis:
1. Mit der ersten SCREEN-Anweisung wird SCALE 1,1 initialisiert.
2. Mit der Anweisung SCALE werden Streckungen und Stauchungen der bei-

den Achsen möglich und damit Vergrößerungen und Verkleinerungen von 
Darstellungen. Möglich wird aber auch eine Veränderung des Koordinaten-
systems bezüglich der Achsenrichtung.

3. SCALE beeinflusst nicht die mit SIZE eingestellten Parameter für die Schrift-
größe bzw. Schriftrichtung.

Beispiele:
1. Voreingestellt ist das Koordinatensystem, das der Auflösung des Plotters ent-

spricht; für die Zeichenfläche gilt 0 ≤ x ≤ 2550, 0 ≤ y ≤ 1800.

0, 1800 2550,1800

     X
0, 0 2550, 0

2. Mit SCALE 5,10 : ZERO(200,75)
ergibt sich das folgende aktuelle Koordinatensystem, und für die Zeichenflä-
che gilt -200 ≤ x ≤ 300, -75 ≤ y ≤ 105.

-200, 105                            Y    0, 105 300, 105

-200,0                                  0, 0 300, 0
   X

-200, -75                                  0, -75 300, -75

3. Eine Verlegung des Koordinatenursprungs von links unten nach links oben 
und ein Ändern der Richtung der y-Achse wird möglich durch 

SCALE 1,-1 : ZER0(0,-1800),
und es gilt dann 0 ≤ x ≤ 2550, 0 ≤ y ≤ 1800.

0, 0 2550, 0
   X

0,1800 2550,1800
 Y

4. Der Koordinatenursprung kann auch außerhalb der Zeichenfläche des Plot-
ters liegen, ist dann aber nicht erreichbar.
Z. B. wird mit den Anweisungen

SCALE 10,5 : ZERO(-150,-100)
ein Koordinatensystem eingestellt, in dem die Punkte auf der Zeichenfläche in 
den Intervallen

150 ≤ x ≤ 405, 100 ≤ y ≤ 460
liegen.

Y                               150,460 405,460

                               150, 100 405,100

    X
0, 0

Positionierung in den Koordinatenursprung

Format:
HOME

Funktion:
Der Zeichenstift  wird zum Koordinatenursprung, dem Punkt (0,0) des aktuellen 
Koordinatensystems, bewegt.

Hinweise:
1. Die Anweisung HOME ist gleichbedeutend mit PSET(0,0),0.
2. Die Anweisung liefert  dann eine Fehlermeldung,  wenn der Koordinatenur-

sprung nicht auf der Zeichenfläche liegt
(vgl. SCALE-Anweisung).

Grundeinstellung des Plotters

Format:
GCLS

Funktion:
Es wird eine Grundeinstellung des Plotters vorgenommen, die der Anweisungs-
folge

SCALE 1,1 : ZERO(0,0) : HOME

entspricht.

6.2. Anweisungen zum Zeichnen

PSET Setzen eines Punktes
LINE Zeichnen einer Linie (Gerade)
CIRCLE Zeichnen eines Kreises

Die folgenden Anweisungen dienen dem Zeichnen von grafischen Darstellungen 
innerhalb der durch SCALE und ZERO festgelegten Zeichenfläche.

Hinweis:
Die Anweisung SCREEN ,1 schaltet die beim KC85 bereits vorhandenen BASIC-
Anweisungen PSET, LINE und CIRCLE auf den Plotter um. Damit ändert sich 
auch die Syntax dieser Anweisungen. Sie erhalten einen ?SN ERROR, wenn mit 
SCREEN der falsche Modus eingestellt ist.

Setzen eines Punktes

Format 1:
PSET(x,y) [, Stift]
x x-Koordinate des Zielpunktes
y y-Koordinate des Zielpunktes
stift Zeichenstift heben/senken

0 - anheben
1 - absenken (zeichnen)
(Standard: stift = 1)

Format 2:
PSET STEP(xrel, yrel) [, stift]
xrel x-Abstand des Zielpunktes vom Startpunkt
yrel y-Abstand des Zielpunktes vom Startpunkt

Funktion:
Setzen eines Punktes (PSET - point set) und damit Positionieren des Zeichenstif-
tes auf die angegebenen Absolutkoordinaten (Format 1) oder relativ bezüglich 
des letzten erreichten Punktes (Format 2).

Hinweis:
Während der Bewegung ist der Zeichenstift  angehoben.  Standardmäßig (ohne 
Angabe von stift) und bei stift = 1 wird der Zeichenstift beim Erreichen des Ziel-
punktes abgesenkt und wieder angehoben (Setzen des Punktes),  bei stift  = 0 
bleibt er angehoben.

Beispiele:

1. 10 FOR I=0 TO 1000 STEP 100

20 PSET(I,I/2)

30 NEXT I

Damit  wird  die  aus  elf  Punkten  bestehende  Punktreihe  von  (0,0)  bis 
(1000,500) gezeichnet.

2. Das gleiche Ergebnis würden folgende Anweisungen liefern:

10 PSET(0,0)

20 FOR I=0 TO 9

30 PSET STEP(100,50)

40 NEXT I

3. Die gleiche Punktreihe in einem anderen Koordinatensystem entsteht mit:

10 SCALE 10,5

20 FOR I=0 TO 100 STEP 10

30 PSET(I,I)

40 NEXT I

In den folgenden Anweisungen wird nur das Format 1
aufgeführt. Alle Koordinatenangaben (x,y)
können auch relativ in der Form STEP(xrel,yrel)
geschrieben werden.

Zeichnen von Linie oder Rechteck

Format:
LINE[(x1,y1)]-(x2,y2)[,[stift][,B]]
x1 x-Koordinate des Anfangspunktes der Linie
y1 y-Koordinate des Anfangspunktes der Linie
x2 x-Koordinate des Endpunktes
y2 y-Koordinate des Endpunktes
stift Zeichenstift heben/senken

0 - anheben
1 - senken (zeichnen)
(Standard: stift = 1)

B - Zeichnen eines Rechteckes (box)

Funktion:
Zeichnen einer Linie oder eines Rechteckes.
1. Sind die Koordinaten (x1,y1) angegeben, so wird der Stift zunächst angeho-

ben und dorthin  positioniert,  ansonsten wird als Anfangspunkt  die  augen-
blickliche Stiftposition angenommen.

2. Ist B nicht angegeben, wird der Zeichenstift entsprechend stift gehoben oder 
gesenkt, und die Position (x2,y2) wird angefahren. Nach Erreichen der Posi-
tion bleibt der Stift in seiner Stellung.

3. Ist  B angegeben und stift  = 1,  so wird ein Rechteck mit  den Eckpunkten 
(x1,y1), (x2,y1), (x2,y2), (x1,y2) gezeichnet und anschließend der Zeichen-
stift auf (x2,y2) positioniert. Gilt stift = 0, so wird der Zeichenstift nur positio-
niert.

Hinweis:
Erfolgt die erste Koordinatenangabe bei LINE relativ, so ist unbedingt zwischen 
LINE und STEP ein Leerzeichen zu schreiben!

LINE STEP(50,60)-(70,80)

Beispiel:
Die  folgenden  Anweisungen  zeichnen  ein  Rechteck  "zwischen"  den  Punkten 
(100,100) und (600,300). Die Stiftbewegung ist durch die Pfeile angedeutet.

10 PSET(100,100),0

20 LINE - STEP(500,200),,B

Das gleiche Rechteck zeichnet man durch

10 LINE(100,100)-STEP(500,200),1,B

oder durch

10 L1NE(100,100)-(600,300),1,B

Zeichnen eines Kreises

Format:
CIRCLE(x,y), radius[,stift [, anf-winkeI [, end-winkel[, ellip]]]]
CIRCLE(x,y), radius,[stift],[anf-winkeI],[end-winkel], ellip
x x-Koordinate des Mittelpunktes
y y-Koordinate des Mittelpunktes
stift Zeichenstift heben/senken

0 - anheben
1 - senken (zeichnen)
(Standard: stift = 1)

anf-winkel Anfangswinkel für Kreisbogen (Bogenmaß) 
end-winkel Endwinkel für Kreisbogen (Bogenmaß) 
ellip Ellipsenparameter (Standard: ellip = 1)

Funktion:
Zeichnen eines Kreises, eines Kreisbogens oder einer Ellipse mit dem angegebe-
nen Radius um den Mittelpunkt (x,y).
1. Die Parameter stift, anf-winkel, end-winkel und ellip können von rechts oder 

einzeln weggelassen werden.
Standard: anf-winkel = 0 

end-winkel = 2 π
Für beide Winkel gilt: -2 π ≤ winkel ≤ 2 π

2. Sind die Winkel negativ angegeben, so wird der Betrag der Winkel gezeich-
net und zusätzlich die entsprechenden Radien (Sektorgrenzen).

3. Nach dem Zeichnen des Kreises bzw. des Kreisbogens und ggf. der Radien 
steht der Zeichenstift im Mittelpunkt (x,y). Für stift = 0 wird der Zeichenstift 
nur dorthin positioniert.

4. Der  Radius  wird,  ebenso  wie  die  Koordinaten,  im  eingestellten  aktuellen 
Koordinatensystem angegeben.

5. Für den Ellipsenparameter gilt
0 < ellip,

er gibt das Verhältnis von y- zu x-Radius an. Die Ellipse wird immer entspre-
chend der Abbildung innerhalb des Kreises mit dem angegebenen Radius 
gezeichnet.

ellip > 1 ellip < 1

ellip>1: der  in  der  CIRCLE-Anweisung angegebene  radius  ist  der  y-
Radius der Ellipse.
x-Radius = radius/ellip

ellip<1: Der angegebene radius ist der x-Radius der Ellipse.
y-Radius = radius*ellip

Beispiele:
1. Die folgende Anweisung zeichnet einen Kreis mit dem Radius 200 (20,0 mm) 

um den Mittelpunkt (500,200):

10 CIRCLE(500,200),200

2. Die angegebene Anweisungsfolge zeichnet ein "Kuchendiagramm".

10 CIRCLE(500,500),200,,-PI/4,-3*PI/2

20 CIRCLE STEP(50,-20),200,,-3*PI/2,-PI/4

      500 550

3. Die nachfolgenden Anweisungen zeichnen einen Kreis und je zwei im Kreis 
liegende Ellipsen, gestaucht in y- bzw. x-Richtung.

10 PSET(300,300),0

20 FOR I=0 TO 4

30 READ EL

40 CIRCLE STEP(0,0),200,,,,EL/15

50 NEXT I

60 HOME

70 DATA 3,9,15,25,75

6.3. Anweisungen zum Schreiben

SIZE Festlegen der Schriftart
LABEL Ausgabe einer Zeichenkette

Mit den Anweisungen SIZE und LABEL wird es Ihnen möglich, Diagramme und 
andere  Zeichnungen  sehr  variabel  zu  beschriften.  Schriftgröße,  -schräge  und 
Schreibrichtung lassen sich nahezu beliebig einstellen.

Festlegung der Schriftart

Format:
SIZE b,l[,r[,s[,a]]]
SIZE b,l,[r],[s],a
b Breite der Buchstaben (Geräteeinheiten)
I Länge (Geräteeinheiten)
r Schreibrichtung (Bogenmaß)
s Schräglage (Bogenmaß)
a Abstand

0 - gleichabständig (Standard)
1 - proportional

Funktion:
Festlegen der Abmessungen der Zeichen (in 0,1 mm) und der Schreibrichtung 
und Schräglage der Zeichen sowie des Zeichenabstandes (vgl. Skizze).

Hinweise:
1. Standardmäßig ist SIZE 24,40,0,0,0 eingestellt. Sind r, s oder a nicht ange-

geben, so gilt für sie der Wert Null.
2. Die Schräglage s gibt die Abweichung der Zeichen von der Senkrechten an 

(bezogen auf die Schreibrichtung).
3. Bei der Ausgabe von Zeichenketten mit proportionalem Abstand (a=1) wird 

die Zeichenbreite bei schmalen Zeichen (I, fast alle Kleinbuchstaben u.a.) um 
0,2*b verringert.

4. Schriftgröße und -richtung werden durch die  Parameter  von SCALE nicht 
beeinflusst.

Ausgabe einer Zeichenkette

Format:
LABEL zeichenkettenausdruck

Funktion:
Ausgabe einer Zeichenkette entsprechend der mit SIZE eingestellten Parameter 
ab aktueller Stiftposition.

Hinweise:
1. Die Stiftposition kann vor der Ausgabe mit PSET(x,y),0 verändert werden.
2. CHR$(127) wird als "ß" ausgegeben (s. Anhang B, Zeichensatz).
3. Ist  CHR$(13)  Bestandteil  der  Zeichenkette,  beginnt  die  Ausgabe  der  auf 

CHR$(13) folgenden Zeichen "auf der nächsten Zeile", d. h. mit einem Zei-
lenabstand entsprechend SIZE unter dem ersten Zeichen der Zeichenkette.

4. Mit einem vorangestellten CHR$(24) können Umlaute dargestellt werden.
5. Mit den Sonderzeichen CHR$(25) bis CHR$(31) können Punkte eines Poly-

gonzugs gekennzeichnet werden (siehe Beispiel 4).

Beispiele:

1. LABEL "ABC"+CHR$(13)+"XYZ"
bewirkt auf dem Plotter die Ausgabe von

2. LABEL CHR$(24)+"A"+CHR$(24)+"a"
bewirkt die Ausgabe von

3. Eine Ausgabe in y-Richtung erreicht man durch

10 SIZE 24,45,PI/2,0
20 LABEL "y-Richtung"

4. Verwendung  der  Sonderzeichen  CHR$(25)  bis  CHR$(28)  zur  Kennzeich-
nung von Punkten eines Polygonzuges:

10 SIZE 24,40
20 HOME
30 DATA 100,100,300,150,500,350,700,50
40 FOR I=0 TO 3
50 READ X,Y
60 LINE-(X,Y):LABEL CHR$(I+25)
70 NEXT I
80 HOME

7. Übersicht der Anhänge

Anhang A Programme

Anhang B Zeichensatz des Kleinplotters XY 4131
- Codierung der ASCII-Zeichen
- Codierung der Sonderzeichen

Anhang C Zeichenfläche und Schriftparameter

Anhang D Speicherbelegung Plotter BASIC 2.0

Anhang E Fehlermeldungen

Anhang F Wartung und Service

7.1. Anhang A: Programme

Für den KC85 stehen derzeit folgende Programme zur Verfügung.

Programmname Kurzbeschreibung Adressen bzw.
Länge in Byte, ca.

PLBASIC.KCC Plottwer BASIC 2.0
BASIC-Erweiterung mit  Plot-
tertreiber

0300 15D1 1420

PLDEMO.SSS Demonstrationsprogramm für 
Kleinplotter XY4131

11 KByte

RASTER.SSS Beispielprogramm zum 
Zeichnen eines karierten 
Blattes

256 Byte

ERICH.SSS Beispiel zur Umsetzung einer 
SVG-Zeichnung zur Plotter-
ausgabe

15 KByte

PLBASIC.KCC

Kurzbezeichnung
BASIC-Erweiterung  "Plotter  BASIC  2.0"  für  die  Kleincomputer  KC 85/3  bis 
KC 85/5 und Treiber für Kleinplotter XY 4131.

Voraussetzungen
- 16K RAM-Modul M022 beim KC85/4 oder KC85/5
- 64K RAM-Modul M011 beim KC85/3
- DIGITAL IN/OUT Modul M001
- Kleinplotter XY 4131 angeschlossen am M001

Inhaltsbeschreibung
PLBASIC.KCC ergänzt den im Grundgerät  vorhandenen BASIC-Interpreter um 
zusätzliche Plotter-Grafik-Befehle und ermöglicht die Nutzung des Kleinplotters 
XY 4131 mit den Kleincomputern KC 85/3 bis KC 85/5.

Hinweise zur Programmnutzung
1. PLBASIC.KCC ist vom CAOS-Menü aus mit %LOAD zu laden und ist selbst-

startend.
2. Der  unter  BASIC  nutzbare  Speicherbereich  ist  durch  diese  Erweiterung 

begrenzt auf maximal 43246 Byte.
3. PLBASIC.KCC kopiert die 4,5KByte große BASIC-Erweiterung mit dem Plot-

ter-Treiber an das obere Speicherende vor C000H.
4. Das  Programm PLBASIC verwendet  den Speicherbereich  von  0234H bis 

0300H als Arbeitsspeicher. Dieser Bereich ist damit für andere Programme 
nicht nutzbar!

PLDEMO.SSS

Kurzbezeichnung
Demonstrations- und Testprogramm zum Kleinplotter XY 4131 am KC 85.

Voraussetzungen
- Plotter BASIC 2.0 aktiv
- Kleinplotter XY 4131 angeschlossen am M001 DIGITAL IN/OUT

Inhaltsbeschreibung
Mithilfe dieses Programms können verschiedene Zeichnungen auf dem Kleinplot-
ter im Format DIN A4 gezeichnet werden. Durch Zeichnen des Bildes 1 "TEST" 
werden dabei alle Plotterfunktionen und Grafikanweisungen angesprochen.

Bild 8: Plotter-Test (1)

Bild 9: Zwiebelmuster 1

Bild 10: Zwiebelmuster 2

Bild 11: Zylinder (4)

Bild 12: Rotierendes Dreieck (5)

Bild 13: Ornament 1

Bild 14: Ornament 2

Bild 15: Ornament 3

Bild 16: Informationsblatt (9)

Bild 17: Technische Zeichnung (10)

Bild 18: DDR-Karte (11)

RASTER.SSS

Kurzbezeichnung
Zeichnen eines karierten Blattes.auf dem Kleinplotter XY 4131 am KC 85.

Voraussetzungen
- Plotter BASIC 2.0 aktiv
- Kleinplotter XY 4131 angeschlossen am M001 DIGITAL IN/OUT

Inhaltsbeschreibung
Das kleine BASIC-Programm zeichnet Linien im Raster von 5 mm zunächst waa-
gerecht und danach senkrecht. Es ist so optimiert, dass der Plotterstift immer im 
Wechsel von einer Papierseite zur anderen und dann wieder in der Gegenrich-
tung zeichnet.

10 SCREEN 0,1

20 FOR Y=0 TO 2500 STEP 100

30 LINE (Y,0)-(Y,1800),1

40 LINE (Y+50,1800)-(Y+50,0),1

50 NEXT Y

60 FOR X=0 TO 1700 STEP 100

70 LINE (2550,X)-(0,X),1

80 LINE (0,X+50)-(2550,X+50),1

90 NEXT X

100 LINE (2550,1800)-(0,1800),1

110 HOME

ERICH.SSS

Kurzbezeichnung
Umsetzung einer SVG-Grafik zur Ausgabe auf dem Kleinplotter XY 4131.

Voraussetzungen
- Plotter BASIC 2.0 aktiv
- Kleinplotter XY 4131 angeschlossen am M001 DIGITAL IN/OUT

Inhaltsbeschreibung
Auf  https://oldcrap.org/2020/03/15/robotron-kc87-plotting-svg-graphics/ wird  die 
Umsetzung einer  SVG-Grafik  zur  Plotterausgabe am KC 87 beschrieben.  Das 
dort vorgestellte BASIC-Programm liest die Plotter-Daten aus dem RAM. In der 
Umsetzung für den KC 85 wurden die Grafikdaten als DATA-Zeilen dem BASIC-
Programm ab der Zeile 1000 direkt eingefügt.

10 REM *******************************

20 REM * DDR Erich Honecker Drawing  *

30 REM *   (c) 2020 oldcrap.org      *

40 REM *******************************

50 REM

60 SCREEN 0,1:GCLS

70 SCALE 2.9,2.9

80 RESTORE

90 START=1:OX=-2:OY=-2

100 READ X,Y

110 IF OX=X AND OY=Y THEN GOTO90

120 OX=X:OY=Y

130 IF X=-1 THEN GOTO180

140 A=Y:Y=X+100:X=A

150 IF START=1 THEN PSET(X,Y)

160 LINE-(X,Y):START=0

170 GOTO100

180 SIZE 30,45,PI/2,0

190 PSET(700,50),0

200 LABEL "* Robotron KC 87 * https://oldcrap.org *"

210 HOME

220 END

1000 DATA 245,0

1001 DATA 238,0

1002 DATA 239,1

1003 DATA ...

7.2. Anhang B: Zeichensatz des Kleinplotters XY 4131

Dezimal Hex. Zeichen Dezimal Hex. Zeichen Dezimal Hex. Zeichen

32 20 Space 64 40 @ 96 60 ˋ
33 21 ! 65 41 A 97 61 a
34 22 " 66 42 B 98 62 b
35 23 # 67 43 C 99 63 c
36 24 $ 68 44 D 100 64 d
37 25 % 69 45 E 101 65 e
38 26 & 70 46 F 102 66 f
39 27 ´ 71 47 G 103 67 g
40 28 ( 72 48 H 104 68 h
41 29 ) 73 49 I 105 69 i
42 2A * 74 4A J 106 6A j
43 2B + 75 4B K 107 6B k
44 2C , 76 4C L 108 6C l
45 2D - 77 4D M 109 6D m
46 2E . 78 4E N 110 6E n
47 2F / 79 4F O 112 6F o
48 30 0 80 50 P 113 70 p
49 31 1 81 51 Q 114 71 q
50 32 2 82 52 R 115 72 r
61 33 3 83 53 S 116 73 s
52 34 4 84 54 T 117 74 t
53 35 5 85 55 U 118 75 u
54 36 6 86 56 V 119 76 v
55 37 7 87 57 W 120 77 w
56 38 8 88 58 X 121 78 x
57 39 9 89 59 Y 122 79 y
58 3A : 90 5A Z 123 7A z
59 3B ; 91 5B [ 124 7B {
60 3C < 92 5C \ 125 7C |
61 3D = 93 5D ] 126 7D }
62 3E > 94 5E ^ 127 7E 
63 3F ? 95 5F _ 128 7F ß

Dezimalcode 24 25 26 27 28 29 30 31

Sonderzeichen " □    + x *

7.3. Anhang C: Zeichenfläche und Schriftparameter

Zeichenfläche des Plotters
(Geräteeinheiten: 0,1 mm)

Schriftparameter
SIZE b, I, r, s, a

b, l in Geräteeinheiten
(0,1 mm)

r, s in Bogenmaß

a Zeichenabstand
0 - gleichmäßig
1 - proportional

Bild 20: Schriftparameter

7.4. Anhang D: Speicherbelegung Plotter BASIC 2.0

RAM 0 Arbeitsspeicher des Betriebssystems 0
Arbeitszellen des Plotter-Treibers 234H
Arbeitsspeicher von BASIC 300H

400H
Anwenderspeicher RAM0
Beginn BASIC-Programm

4000H
RAM 4 RAM4 beim KC85/4 bzw. KC85/5

RAM-Erweiterungsmodul M011 beim KC85/3

RAM 8 RAM8 beim KC85/4 bzw. KC85/5 8000H
RAM-Erweiterungsmodul M011 beim KC85/3

Plotter-Treiber und BASIC-Erweiterung AE00H
(4,5 KByte)

RAM C RAM-Erweiterungsmodul M022 oder M011 C000H
Kopie des BASIC-ROM mit Modifikationen
(8 KByte)

ROM E CAOS-Betriebssystem E000H
(8 KByte)

FFFFH

7.5. Anhang E: Fehlermeldungen

Folgende Fehlermeldungen können bei der Nutzung des Kleinplotters XY 4131 
mit Plotter BASIC 2.0 zusätzlich zu den Meldungen des BASIC-Interpreters (vgl. 
Programmierhandbuch, Anhang H) auftreten:

?SN ERROR Treiberprogramm für Plotter PLBASIC.KCC ist nicht 
geladen bzw. nicht verfügbar?

?OV ERROR Plotter wurde nicht ordnungsgemäß mit SCREEN 
initialisiert

?PO ERROR
(Plotter an error)

Papierandruckhebel am Plotter nicht geschlossen 
oder Plotter mit SCREEN nicht zugewiesen

X OUT OF RANGE x-Koordinate ) bei Stiftbewegung
Y OUT OF RANGE y-Koordinate ) außerhalb der
XY OUT OF RANGE x- und y-Koordinaten ) Zeichenfläche

7.6. Anhang F: Wartung und Service

Der Kleinplotter XY 4131 arbeitet zuverlässig und wartungsfrei.  Es wird jedoch 
empfohlen, das Gerät bei Nichtgebrauch mit der beiliegenden Schutzhülle abzu-
decken und in bestimmten Zeitabständen mit einem weichen Lappen oder Staub-
pinsel zu reinigen. Bei stärkerer Verschmutzung kann ein feuchtes Tuch oder Spi-
ritus verwendet werden.

Trotzdem können beim Betreiben Störungen auftreten. Eine der häufigsten ist 
das Einreißen des Papiers durch die Schreibstiftspitze. Als Ursache dafür kom-
men zu dünnes Papier oder zu spitze Zeichenstifte in Betracht. Der Zeichenstift 
kann auch zu niedrig über der Zeichenfläche positioniert sein, sodass der Druck 
der Zeichenspitze auf das Papier zu groß ist. Bei dadurch hervorgerufenen und 
ähnlichen Störungen sollten Sie immer folgende Reihenfolge der Bedienhandlun-
gen einhalten:

1. Papierhalterung öffnen (Hebel 9 nach oben kippen), der Zeichenstift fährt zur 
Grundstellung (entspricht GCLS)

2. Fehlermeldung des Computers abwarten
?PO ERROR

3. Störung beseitigen
- neues Papier einlegen, Hebel 9 nach unten,
- Stift neu positionieren

4. Programm erneut starten
(SCREEN-Anweisung muss abgearbeitet werden), fertig.

Wenn Sie den Plotter beim Auftreten einer Störung mit dem Netzschalter (10) 
außer Betrieb nehmen, so müssen Sie folgendes beachten:

5. Fehlermeldung des Computers abwarten
?PO ERROR

6. Plotter einschalten
7. Weiter bei 3.

Bei anderen Reihenfolgen der Bedienhandlungen sind die notwendige Neuinitiali-
sierung des Plottertreibers und damit  eine ordnungsgemäße Weiterarbeit  nicht 
gesichert.

Die Servicewerkstätten entnehmen Sie bitte dem beiliegenden Informationsblatt.

7.7. BASIC-Kurzbeschreibung zum KLEINPLOTTER XY 4131

Zeichenfläche 255  mm  x  180  mm  =  2550  x  1800  Punkte 
(Abstand 0,1 nm)

Zuweisung des Plotters
SCREEN 0,1 Grafikanweisungen wirken auf den Plotter am 

Plotteranschluss des Moduls M001;
Initialisierung (SCALE 1,1)

Koordinatensysteme und Grundeinstellungen
ZERO(X,Y) Festlegung  des  Nullpunktes  des  aktuellen 

Koordinatensystems in den Punkt (X,Y)
SCALE XF,YF Maßstabsfestlegung in x- bzw. y-Richtung (alle 

Koordinatenangaben werden vor der Stiftbewe-
gung mit XF bzw. YF multipliziert)

HOME Stiftbewegung  zum  Punkt(0,0)  des  aktuellen 
Koordinatensystems = PSET(0,0),0

GCLS Grundeinstellung  des  Plotters;  entspricht 
ZER0(0,0):HOME

Anweisungen zum Zeichnen
Die Koordinaten können "absolut" durch (X,Y)
oder "relativ bezüglich des letzten Punktes" durch STEP(X,Y)
angegeben werden.

PSET(X,Y),0 Stift zum Punkt (X,Y) bewegen
PSET STEP(X,Y),1 Stift um Abstand (X,Y) verschieben und Punkt 

setzen
LINE(X,Y)-(U,V) Linie (Gerade) zwischen (X,Y) und (U,V) zeich-

nen
LINE-(U,V) Linie zum Punkt (U,V) ziehen
LINE(X,Y)-(U,V),1,B Rechteck  mit  den  Eckpunkten  (X,Y),  (U,Y), 

(U,V), (X,V) zeichnen.
LINE STEP(X,Y)-STEP(U,V) Linie zwischen den relativ auf die letzte Stiftpo-

sition  bezogenen  Punkten  (X,Y)  und  (U,V) 
zeichnen. (Zwischen LINE und STEP muss ein 
Leerzeichen gesetzt werden!)

CIRCLE(X,Y),R Kreis  um den Mittelpunkt  (X,Y)  mit  Radius R 
zeichnen.

CIRCLE(X,Y),R,1,PI/4,PI,2 Ellipse um den Mittelpunkt (X,Y) zeichnen. Der 
Ellipsenfaktor ist 2 (Stauchung in X-Richtung). 
Es wird nur der Ellipsenbogen zwischen P1/4 
und PI gezeichnet.

Anweisungen zur Schriftausgabe
SIZE 30,50,PI/2,PI/12,1 Festlegung der Schriftart:

Breite b = 10 (3 mm)
Länge l = 50 (5 mm)
Schriftrichtung r = PI/2 (y-Richtung)
Schräglage s = PI/12 (Bogenmaß)
Zeichenabstand = proportional

SIZE 24,40,0,0,0 Standardschrift in x-Richtung
LABEL T$ Die Zeichenkette T$ wird ab der aktuellen Stift-

position  mit  den  gültigen  Schriftparametern 
ausgegeben.

